# Parole media player
#

pkgname=parole
pkgver=0.2.0.6
pkgrel=1
pkgdesc="GStreamer based media player for Xfce desktop"
arch=('ppc')
url="http://goodies.xfce.org/projects/applications/parole/"
license=('GPL2')
depends=('gtk2' 'glib2' 'dbus-glib' 'gstreamer0.10-base' \
	'libnotify' 'libxfcegui4' 'libxfce4util' 'taglib' \
	'libsm' 'libx11' 'hicolor-icon-theme' \
	'desktop-file-utils')
makedepends=('intltool' 'pkg-config' 'gettext')
options=('!libtool')
install=${pkgname}.install
source=(http://archive.xfce.org/src/apps/${pkgname}/0.2/${pkgname}-${pkgver}.tar.bz2
	configure.patch)
md5sums=('5d7e5ab7535bdf89508a8d7e551f9187'
	'aa5d59031a8e4e4452a8e03e85adcf87')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	#
	patch -Np0 -i ${srcdir}/configure.patch

	sh configure --prefix=/usr \
		--sysconfdir=/etc \
		--libexecdir=/usr/lib/xfce4 \
		--localstatedir=/var \
		--disable-static \
		--enable-nls \
		--enable-libnotify \
		--enable-taglib \
		--enable-properties-plugin \
		--enable-tray-plugin \
		--enable-window-title-plugin \
		--enable-power-manager-plugin \
		--disable-gtk-doc-html \
		--without-html-dir

	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}